



MR. SALTY

Crop Yield Reductions Due to Soil Salinity

Version 1.1
for the
IBM Personal Computer

NORTH DAKOTA
STATE UNIVERSITY
MAR 24 1986
SERIALS DEPT.
LIBRARY

By

E.H. Vasey
Extension Soils Specialist


Carl D. Fanning
Extension Soils Specialist

Alex Maianu
Associate Professor of Soil Science

Mike Vasey and Rick Koon
Student Computer Programmers

Plant Science Section
North Dakota Cooperative Extension Service
North Dakota State University

S
544.3
N9
98x
10.873



North Dakota State University Fargo, North Dakota 58105

Introduction

MR. SALTY is a prediction program for crop yield reduction due to soil salinity (soluble salts). The program was developed by Dr. E.H. Vasey using one year's data collected in North Dakota by Dr. Alex Maianu.

There are over 1 million acres of saline soils in North Dakota.(3) These occur in all parts of the state. Most of the saline acres in North Dakota have been in existence for many years. Saline soil areas caused by seepage have been a growing concern in more recent years. (1,2,5)

The reclamation of highly saline soils usually is not practical under dryland farming conditions. (3,4) These soils probably are best suited for grassland use. Cost inputs toward reclamation hardly ever give paying returns on these types of saline soils.

Better management of slightly to moderately saline soils for dryland field crop production can improve yields, thus increase profits or at least reduce cost inputs for producers.

User's Guide

To use MR. SALTY, when the computer is off place the diskette into drive A and turn on your system. If your system is on, insert the diskette into drive A and re-boot the system by simultaneously depressing the Ctrl and Alt keys, then press the Del key. The program will automatically load and run.

Screen 1 in MR. SALTY is introductory remarks. The screen provides information on what crops the program has data for, where the data for the prediction equations came from and a general outline of other information available in the program. Pressing any white-colored key will move the program to screen 2.

Screen 2 begins by asking for a field identification. This should be some name or symbol so that you can identify the field where the soil salinity readings or estimates were taken. Soil salinity estimates can be made using data contained in soil survey publications. The field identification can be any combination of letter, numbers and punctuation marks you desire up to 25 characters in length. If you try to enter more than 25 characters, the program will beep and truncate or cut off the attempted entry at 25 characters. After entering the field identification, the screen will prompt you for a soil salinity reading for the designated field. This reading should be obtained by the common laboratory method for estimating the soluble salt content of the soil, which is determining the electrical conductivity of a 1:1 soil-water suspension. This is the method used at the soil testing laboratory at North Dakota State

University. You can see this information plus some general soil salinity guidelines in the HELP function ([F1] key to this prompt.

Following input of the soil salinity reading or estimate, screen 2 will present the seven crops you presently have the option of using in the program. You may elect to see only one crop or may compare all seven crops and the yield reduction due to soil salinity on each.

To select a crop, enter its corresponding number from 1 to 7 and press the enter key. You will then be prompted for your normal yield of that crop. A check mark will appear to the left of the crop chosen to indicate it has been selected. When you have entered the desired crop or crops, you may see the results by striking the [F3] key. If you select all seven crops, the program will automatically move to screen 3. Additionally, you may abort (return to DOS) the program at any time during this prompt by depressing the [F2] key or use the HELP function by depressing the [F1] key.

Screen 3 will present a graphic display of the percent yield reduction you can expect at the selected salinity level in the given field. The left hand side of the graph will present the name of the crop and the expected normal yield you inputted into the program. Line 25 will display four options:

- [F1] Continue Program - moves the program to the next series of screens and prompts.
- [F2] Printer - sends the data to the printer. The data will not be in graphic form; you may obtain a copy of the screen in graphic form by using the PC's PrtSc (Print Screen) function.
- [F3] Stop - abort the program and return to DOS.
- [F4] More Crops - return to screen 2 and repeat the process as described in the paragraph above.

Electing to continue the program ([F1]) will present a prompt on guideline information on if cropping attempts are likely to be profitable. Answering "Y" will display the information. Responding "N" will present the next prompt concerning information on saline soil management. Responding "Y" to the saline soils management prompt will present Summary Comments and a prompt asking if you would like further details on any of the comments. If you respond "Y," you will be asked if you want a printed copy of comments. You must then enter "Y" or "N." You will then be prompted to give the number of the comment you desire further information on (a check mark will appear by the comment(s) selected). Striking the enter key will end this prompt and display the requested comments. If you plan to select more than four expanded comments then you should probably send the comments to the printer since the first comments may scroll off the screen. Responding "N" to the prompt regarding Summary Comments

will move the program onto the next screen pertaining to sources of additional information.

Continuing, you will be prompted to see a listing of sources of additional information. These are publications available to you to further assist in management of soil salinity problems.

The final prompt asks if you would like to see soil salinity figures for another field. Responding "Y" will return you to the field identification prompt (screen 2) and the program repeats itself from there, responding "N" returns you to DOS.

REFERENCES

1. Dodds, Duaine, 1974, "Forages for Salt-affected and Wet Soils." Extension Circular R-584, North Dakota Cooperative Extension Service, North Dakota State University, Fargo, ND.


2. Holm, H.M. and J.L. Henry, "Understanding Salt-affected Soils," Saskatchewan Agriculture, Plant Industry Branch.

3. Johnsgard, Gordon A., 1974, "Salt Affected Problem Soils in North Dakota," Extension Bulletin No. 2, Cooperative Extension Service, North Dakota State University, Fargo, ND.

4. Omodt, Hollis W., Fred W. Schroer and Donald D. Patterson, 1975, "The Properties of Important Agricultural Soils as Criteria for Mined Land Reclamation," Bulletin 492, Department of Soils, Agricultural Experiment Station, North Dakota State University, Fargo, ND.

5. Sommerfeldt, T.G. and E. Rapp, 1982, "Management of Saline Soils," Agriculture Canada publication 1674/E, Communications Branch, Agriculture Canada, Ottawa.

North Dakota State University and the program authors shall have no liability or responsibility to any person or entity with respect to any liability, loss or damage caused or alleged to be caused directly or indirectly by this software, including, but not limited to any interruption of service, loss of business or anticipatory profits or consequential damages resulting from the use or operation of this software.



Data Dictionary For Mr. Salty

Cropname\$	- a seven cell array that holds the name of the crop selected.	The following variables are used by the screen in routine. For further details see Advanced BASIC and Beyond for the IBM/PC by Dr. Larry J. Goldstein.	
Ryld	- relative yield, a seven cell array that holds the percent relative yield of crops selected.	caller	- identifies a set of acceptable characters
Nrmlyld	- normal yield, a seven cell array to hold the user input normal yield for crops selected.	minkey	- the lowest value character acceptable
Fallyld	- final yield, a seven cell array to hold the yield a user can expect after the percent yield reduction has been applied to Nrmlyld.	maxkey	- the highest value character acceptable
FKey\$, FKey1\$, FKey2\$	- string variables that will display varied function key commands for user.	extminkey	- the lowest value extended character acceptable
ST\$	- a string\$ variable holding 79 blanks. Used to erase lines throughout the program.	extmaxkey	- the highest value extended character acceptable
FieldID\$	- the field identification the user wants to give the field the calculations will be based on. Can be any combination of letters or numbers up to 25.	special key\$	- acceptable special keys
SS, SS\$	- soil salinity	extspecial key\$	- acceptable extended special keys
The following variables are workers internal to the program.		xfld	- column position where output is to begin
X%, I%	- used in for-next loops	yfld	- row where output is to be displayed
true, false	- boolean values used as flags	Lngth	- maximum number of characters allowed
num%	- counter to hold the number of crops selected	A description of other variable is contained in the program in the screen in routine.	
SS\$	- holds responses to various yes/no questions		
row%	- the row number on screen		
col%	- the column number on screen		
Lp%	- a boolean value to activate the printer		
b1, b2, k1, k2	- program constants used to determine ss		
cmt\$	- used to read/write information from data statements		

```

5  ^*****
10 ^*                MR. SALTY                VERSION 1.1                *
15 ^*
20 ^*  written by:  michael w. vasey                *
25 ^*  revised by:  ricky m. koon, 5-84, second revision 7-84                *
30 ^*
35 ^*****
40 ON ERROR GOTO 2955: SCREEN 0: KEY OFF: WIDTH 80: CLS
45 CR$ = CHR$(17)+CHR$(196)+CHR$(217)
50 DEF SEG ^-----> clear the keyboard buffer
55 ^-----> dimension variables for the seven different crops
60 DIM CROPNAME$(7), RYLD(7), NRMLYLD(7), FINALYLD(7)
65 DIM MINKEY(5), MAXKEY(5), EXTMINKEY(5), EXTMAXKEY(5), SPECIALKEY$(5),
    EXTSPECIALKEY$(5), CAPSON(5)
70 FKEY$ = "[F1] Continue Program [F2] Printer [F3] Stop [F4] More Crops"
75 FKEY1$ = "[F1] HELP [F2] STOP"
80 FKEY2$ = "[F1] HELP [F2] STOP [F3] DISPLAY RESULTS"
85 FOR X% = 1 TO 10 : KEY X%,"": NEXT X% ^-----> disable the soft keys
90 TRUE = -1 : FALSE = 0
95 ^-----> initialize keys
100 ^-----> text input caller = 1
105 MINKEY(1) = 32: MAXKEY(1) = 127: EXTMINKEY(1)= 59: EXTMAXKEY(1) = 60:
    SPECIALKEY$(1) = CHR$(8) + CHR$(13) + CHR$(27):EXTSPECIALKEY$(1) =
    "":CAPSON(1) = -1
110 ^-----> numeric input caller = 2
115 MINKEY(2) = 46: MAXKEY(2) = 57: EXTMINKEY(2) = 59: EXTMAXKEY(2) = 61:
    SPECIALKEY$(2) = CHR$(8) + CHR$(13) + CHR$(27) + "0123456789": CAPSON(2) = 0
120 EXTSPECIALKEY$(2) = ""
125 ^-----> text input caller = 3
130 MINKEY(3) = 48: MAXKEY(3) = 57: EXTMINKEY(3) = 59: EXTMAXKEY(3) = 61:
    SPECIALKEY$(3) = CHR$(8) + CHR$(13) + CHR$(27): CAPSON(3) = 0:
    EXTSPECIALKEY$(3) = ""
135 ^-----> caller = 4 function keys only
140 MINKEY(4) = 0: MAXKEY(4) = 0: EXTMINKEY(4) = 59: EXTMAXKEY(4) = 62:
    SPECIALKEY$(4) = "": EXTSPECIALKEY$(4) = ""
145 ^-----> caller = 5 Y/N responses
150 MINKEY(5) = 0: MAXKEY(5) = 0: EXTMINKEY(5) = 0: EXTMAXKEY(5) = 0:
    SPECIALKEY$(5) = CHR$(8)+CHR$(13)+CHR$(78)+CHR$(89)+CHR$(110)+CHR$(121)
    :EXTSPECIALKEY$(5) = "":CAPSON(5) = -1
155 ^-----> print opening comments
160 LOCATE 1,28: PRINT "WELCOME TO MR. SALTY"
165 LOCATE 3,12: PRINT "MR. SALTY WAS DEVELOPED BY DR. E.H. VASEY, EXTENSION"
170 PRINT TAB(12) "SOILS SPECIALIST AT NORTH DAKOTA STATE UNIVERSITY,
    USING DATA"
175 PRINT TAB(12) "COLLECTED BY DR. ALEX MAIANU, NDSU SOIL SCIENCE DEPARTMENT."
180 PRINT: PRINT TAB(12) "MR. SALTY IS DESIGNED TO PREDICT YIELD REDUCTION
    DUE TO SOIL"
185 PRINT TAB(12) "SALINITY (SOLUBLE SALTS) IN HARD RED SPRING WHEAT, DURUM"
190 PRINT TAB(12) "WHEAT, BARLEY, FLAX, SUNFLOWER, CORN (GRAIN),"
195 PRINT TAB(12) "AND CORN(SILAGE) WHEN GROWN UNDER NORTH DAKOTA CONDITIONS."
200 PRINT: PRINT
205 PRINT TAB(12) "MR. SALTY WILL ALSO GIVE YOU A GENERAL GUIDELINE AS TO"
210 PRINT TAB(12) "WHEN CROPPING ATTEMPTS ON SALINE SOIL ARE NOT LIKELY TO"
215 PRINT TAB(12) "COVER PRODUCTION COSTS. GENERAL MANAGEMENT SUGGESTIONS"
220 PRINT TAB(12) "FOR SALINE SOIL AND SOURCES OF ADDITIONAL INFORMATION ARE"

```

```

225 PRINT TAB(12) "ALSO AVAILABLE IN THE PROGRAM."
230 PRINT: PRINT
235 PRINT TAB(12) "HELP STATEMENTS APPEAR THROUGHOUT THE PROGRAM AND"
240 PRINT TAB(12) "CAN BE USED BY PRESSING THE APPROPRIATE FUNCTION KEY."
245 PRINT:PRINT SPC(27);:COLOR 0,7:PRINT " press any key to continue":COLOR 7,0
250 A$=INKEY$: IF A$ = "" THEN 250
255 FOR X% = 1 TO 7: RYLD(X%) = 0: CROPNAME$(X%) = "": NRMLYLD(X%) = 0:
    FINALYLD(X%) = 0 :NEXT X%      ^-----> initialize the variables
260 ST$ = STRING$(79,32)
265 CLS:LOCATE 1,10: PRINT "Field Identification:";
270 LOCATE 25,10: PRINT FKEY1$;
275 CALLER = 1: XFLD = 32: YFLD = 1: LNGTH = 26
280 GOSUB 2290
285 IF E$ <> "" THEN 290 ELSE 305
290 C = ASC(E$): EXTENDED = FALSE
295 IF C = 59 THEN GOSUB 1635: GOTO 265
300 IF C = 60 THEN GOTO 815
305 FIELDID$ = "~" + S$ + "~"
310 LOCATE 3,10: PRINT "What is the salinity (soluble salt) reading for field"
315 PRINT TAB(10) FIELDID$;" in the 0 to 6 inch depth? ";
320 CALLER = 2: XFLD = POS(0)+1: YFLD = 4: LNGTH = 6
325 GOSUB 2290
330 SS$ = S$: SS = VAL(SS$)
335 IF SS < 0 OR SS > 10 THEN BEEP: GOTO 310
340 SS = SS * 1000
345 IF E$ <> "" THEN 350 ELSE 365
350 C = ASC(C$): EXTENDED = FALSE
355 IF C = 59 THEN GOSUB 1520: GOTO 310
360 IF C = 60 THEN 815
365 LOCATE 25,10: PRINT FKEY2$;:NUM% = 1
370 LOCATE 6,10:PRINT "You may see salinity effects on any or all of the
    following:"
375 LOCATE 8,12: PRINT "1. Hard Red Spring Wheat"
380 LOCATE 9,12: PRINT "2. Durum Wheat"
385 LOCATE 10,12:PRINT "3. Barley"
390 LOCATE 11,12:PRINT "4. Flax"
395 LOCATE 12,12:PRINT "5. Sunflowers"
400 LOCATE 13,12:PRINT "6. Corn (grain)"
405 LOCATE 14,12:PRINT "7. Corn (silage)"
410 LOCATE 20,1:PRINT ST$;:LOCATE 20,10:PRINT "ENTER YOUR CHOICE OF CROP ";
415 CALLER = 2: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
420 GOSUB 2380
425 IF E$ <> "" THEN 430 ELSE 450
430 C = ASC(E$): EXTENDED = FALSE
435 IF C = 59 THEN GOSUB 1675 : GOTO 410
440 IF C = 60 THEN 815
445 IF C = 61 THEN 525
450 CHOICE% = VAL(S$)
455 IF CHOICE% < 1 OR CHOICE% > 7 THEN BEEP: GOTO 410
460 ON CHOICE% GOSUB 845,885,925,965,1005,1045,1085 ^---> execute equation
465 LOCATE 20,1: PRINT ST$
470 LOCATE 20,10: PRINT "Enter your normal yield for ";CROPNAME$(NUM%);
475 CALLER = 3: XFLD = 52: YFLD = 20: LNGTH = 5
480 GOSUB 2290
485 IF E$ <> "" THEN 490 ELSE 510
490 C = ASC(E$): EXTENDED = FALSE

```

```

495 IF C = 59 THEN GOSUB 1725: GOTO 465
500 IF C = 60 THEN 815
505 IF C = 61 THEN 525
510 NRMLYLD(NUM%) = VAL(S$):RLTVYLD = INT(RLTVYLD): RYLD(NUM%) = RLTVYLD
515 FINALLYLD(NUM%) = (RLTVYLD/100) * NRMLYLD(NUM%)
520 NUM% = NUM% + 1: IF NUM% > 7 THEN 525 ELSE 410
525 GOSUB 1380 '-----> print the graph and associated data
530 LOCATE 25,1,0: PRINT ST$;:LOCATE 25,7,0: PRINT FKEY$;
535 CALLER = 4: XFLD = POS(0): YFLD = CSRLIN: LNGTH = 1
540 GOSUB 2380 '-----> screenin routine function keys only
545 C = ASC(E$): EXTENDED = FALSE
550 IF C=59 THEN 570
555 IF C=60 THEN 1890
560 IF C=61 THEN 815
565 IF C=62 THEN CLS: GOTO 365
570 CLS: LOCATE 3,12:PRINT "Would you like to see guideline information on whether"
575 PRINT TAB(12) "cropping attempts on your saline soil may be profitable?(Y/N)";
580 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
585 GOSUB 2380
590 IF S$ = "N" OR S$ = "n" THEN 625
595 CLS: PRINT: PRINT: PRINT TAB(12) "In general, anytime percent of normal
    yield is 50 or less,"
600 PRINT TAB(7) "the yield obtained may not cover the costs of production."
605 PRINT TAB(12) "One should calculate this using your own costs of production"
610 PRINT TAB(7) "expected to be incurred on the area under consideration."
615 LOCATE 23,21: PRINT "press any key to continue"
620 A$ = INKEY$: IF A$ = "" THEN 620
625 CLS
630 PRINT : PRINT
635 PRINT TAB(5) "Would you like to see more information on saline soil
    management (Y/N)?"
640 CALLER = 5:XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
645 GOSUB 2380
650 IF S$ = "N" OR S$ = "n" THEN 720
655 '-----> Summary Comment print routine
660 '
665 CLS: LOCATE 2,1
670 PRINT TAB(12) "SUMMARY COMMENTS":PRINT
675 READ CMT$
680 WHILE CMT$ <> "quit"
685   PRINT TAB(12) CMT$
690   READ CMT$
695 WEND
700 LOCATE 20,7: PRINT "Would you like detailed information on any of
    the above comments(Y/N)?"
705 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
710 GOSUB 2380
715 IF S$ = "Y" OR S$ = "y" THEN GOSUB 1755
720 CLS:PRINT:PRINT TAB(11) "Would you like to see a listing of sources
    of additional";
725 PRINT TAB(11) "information on salt affected soil? (Y/N)";
730 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
735 GOSUB 2380
740 IF S$ = "Y" OR S$ = "y" THEN CLS ELSE GOTO 790
745 '-----> Additional sources print routine
755 READ CMT$

```

```

760 WHILE CMT$ <> "aquit"
765 IF CMT$ <> "*" THEN 780
770   READ CMT$
775   PRINT TAB(12) CMT$
780   READ CMT$
785 WEND
786 '-----> prompt user for another estimate
790 LOCATE 24,12: PRINT "Estimate another field(Y/N)?";
795 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LENGTH = 2
800 GOSUB 2380
805 IF S$ = "N" OR S$ = "n" THEN CLS: SYSTEM
810 CLS:RESTORE: GOTO 255
815 CLS: SYSTEM
820 '*****
825 '*   The constants b1, b2, k1 and k2 are given the appropriate *
830 '* values for the specified crop. These constants were derived *
835 '* using a form of regression analysis. *
840 '*****
845 '*** SPRING WHEAT SUBROUTINE ***
850 CROPNAME$(NUM%) = "Spring Wheat": LOCATE 8,10: PRINT CHR$(251);
855 IF SS < 1725 THEN SS = 1725
860 IF SS > 6400 THEN SS = 6400
865 B1 = -.023201 : B2 = 2.17233E-07
870 K1 = 1800 : K2 = 6400
875 GOSUB 1125
880 RETURN
885 '*** DURUM WHEAT SUBROUTINE ***
890 CROPNAME$(NUM%) = "Durum Wheat": LOCATE 9,10: PRINT CHR$(251);
895 IF SS < 2075 THEN SS = 2075
900 IF SS > 6000 THEN SS = 6000
905 B1 = -.04421 : B2 = 2.2925E-06
910 K1 = 2100 : K2 = 6000
915 GOSUB 1125
920 RETURN
925 '*** BARLEY SUBROUTINE ***
930 CROPNAME$(NUM%) = "Barley":LOCATE 10,10: PRINT CHR$(251);
935 IF SS < 2400 THEN SS = 2400
940 IF SS > 6400 THEN SS = 6400
945 B1 = -.030137 : B2 = 5.83699E-07
950 K1 = 2400 : K2 = 6400
955 GOSUB 1125
960 RETURN
965 '*** FLAX SUBROUTINE ***
970 CROPNAME$(NUM%) = "Flax":LOCATE 11,10: PRINT CHR$(251);
975 IF SS < 2400 THEN SS = 2400
980 IF SS > 6000 THEN SS = 6000
985 B1 = -.054968 : B2 = 3.22395E-06
990 K1 = 2400 : K2 = 6000
995 GOSUB 1125
1000 RETURN
1005 '*** SUNFLOWER SUBROUTINE ***
1010 CROPNAME$(NUM%) = "Sunflower":LOCATE 12,10: PRINT CHR$(251);
1015 IF SS < 1600 THEN SS = 1600
1020 IF SS > 6000 THEN SS = 6000
1025 B1 = -2.02238E-03 : B2 = -2.7375E-06
1030 K1 = 1600 : K2 = 6000

```



```

1035 GOSUB 1125
1040 RETURN
1045 ^*** CORN GRAIN SUBROUTINE ***
1050 CROPNAME$(NUM%) = "Corn(grain)":LOCATE 13,10:PRINT CHR$(251);
1055 IF SS < 600 THEN SS = 600
1060 IF SS > 6000 THEN SS = 6000
1065 B1 = -.038624 : B2 = 3.04628E-06
1070 K1 = 600 : K2 = 6000
1075 GOSUB 1125
1080 RETURN
1085 ^*** CORN SILAGE SUBROUTINE ***
1090 CROPNAME$(NUM%) = "Corn(silage)":LOCATE 14,10: PRINT CHR$(251);
1095 IF SS < 600 THEN SS = 600
1100 IF SS > 6000 THEN SS = 6000
1105 B1 = -.010826 : B2 = -1.166E-06
1110 K1 = 600 : K2 = 6000
1115 GOSUB 1125
1120 RETURN
1125 ^*** EQUATION ***
1130 BK = -(B1 * K2) - (B2 * (K2*K2))
1135 RLTVYLD = BK + (B1 * SS) + (B2 * (SS*SS))
1140 RETURN
1145 END
1150 ^ *****
1155 ^ *                SUMMARY COMMENTS                *
1160 ^ *  comments may be added or deleted from the list.  REMEMBER you  *
1165 ^ *  MUST place the word QUIT at the end of the data!          *
1170 ^ *****
1175 DATA "1. Identify source of salinity."
1180 DATA "2. Improve surface and/or subsurface drainage."
1185 DATA "3. Use deep rooted crops."
1190 DATA "4. Grow salt-tolerant crops."
1195 DATA "5. Seed when moisture is good."
1200 DATA "6. Use sweet clover rotation."
1205 DATA "7. Apply manure."
1210 DATA "8. Use recommended fertilization."
1215 DATA "9. Avoid summerfallow."
1220 DATA "10. Do tillage at proper moisture content."
1225 DATA "11. Chemical amendments won't neutralize salts."
1230 DATA "quit"
1235 ^ *****
1240 ^* The following are sources of additional information.          *
1245 ^* Additional sources may be added to the listing by following the *
1250 ^* same format. BE SURE TO PUT THE AQUIT STATEMENT AT THE END OF  *
1255 ^* OF ALL YOUR SOURCES!                                          *
1260 ^ *****
1265 DATA "*","1. `Salt Affected Problem Soils in North Dakota` "
1270 DATA "*"," Extension Bulletin No. 2"
1275 DATA "*"," available at County Extension Offices in N.D."
1280 DATA "*"," "
1285 DATA "*","2. `The Properties of Important Agricultural Soils as"
1290 DATA "*"," Criteria for Mined Land Reclamation` "
1295 DATA "*"," Experiment Station Bulletin 492"
1300 DATA "*"," available through County Extension Offices in N.D."
1305 DATA "*"," "
1310 DATA "*","3. `Understanding Salt-Affected Soils` "

```

```

1315 DATA "*" ," H. M. Holm and J. L. Henry"
1320 DATA "*" ," available in N.D. from Dr. E. H. Vasey, Cooperative"
1325 DATA "*" ," Extension Service, NDSU, Fargo, ND (701) 237-8883"
1330 DATA "*" ," "
1335 DATA "*" ,"4. `Forages for Salt-Affected and Wet Soils"
1340 DATA "*" ," Extension Circular R-584"
1345 DATA "*" ," available at County Extension Offices in ND"
1350 DATA "*" ," "
1355 DATA "*" ,"5. `Management of Saline Soils ` "
1360 DATA "*" ," T.G. Sommerfeldt and E. Rapp; Publication 1624/E"
1365 DATA "*" ," available from Communications Branch, Agriculture Canada"
1370 DATA "*" ," Ottawa KIA OC7"
1375 DATA "aquit"
1380 ^-----> generate bar chart display
1385 CLS: WIDTH 80: SCREEN 0
1390 LOCATE 3,18: PRINT"0 10 20 30 40 50 60 70
80 90 100"
1395 LOCATE 4,15: PRINT "ENY";:LOCATE 4,18: PRINT STRING$(60,205);
1400 LOCATE 21,18: PRINT STRING$(60,205);
1405 FOR ROW% = 5 TO 20
1410 LOCATE ROW%,18: PRINT CHR$(186);
1415 LOCATE ROW%,78: PRINT CHR$(186);
1420 NEXT ROW%
1425 LOCATE 4,18: PRINT CHR$(201);
1430 LOCATE 4,78: PRINT CHR$(187);
1435 LOCATE 21,18: PRINT CHR$(200);
1440 LOCATE 21,78: PRINT CHR$(188);
1445 FOR COL% = 24 TO 72 STEP 6
1450 FOR ROW% = 5 TO 20
1455 LOCATE ROW%,COL%: PRINT CHR$(250)
1460 NEXT ROW%,COL%
1465 ^-----> display title, cropnames and bars
1470 LOCATE 1,6: PRINT "SOIL SALINITY IS ";SS$;" MMHOS/cm IN FIELD ";FIELDID$
1475 FOR X% = 1 TO 7
1480 ROW% = 3 + (X% * 2)
1485 IF CROPNAME$(X%) = "" THEN 1510
1490 LOCATE ROW%,1: PRINT CROPNAME$(X%);
1495 LOCATE ROW%,14: PRINT USING "####";NRMLYLD(X%);
1500 LOCATE ROW%,19: IF RYLD(X%) >= 1 THEN PRINT STRING$(((RYLD(X%)*.6)-1),219);
1505 NEXT X%
1510 LOCATE 22,18: PRINT CHR$(219) " = percent relative yield" SPC(5)
"ENY = expected normal yield"
1515 RETURN
1520 ^-----> help for salinity figures
1525 CLS: LOCATE 3,12,0: PRINT "The common laboratory method for estimating
the soluble salt"
1530 PRINT TAB(7) "content of soils is detemining the electrical
conductivity of a 1:1"
1535 PRINT TAB(7) "soil - water suspension. The soil testing laboratory
at North Dakota"
1540 PRINT TAB(7) "State University uses this method. The range in
reading values is 0.00"
1545 PRINT TAB(7) "to 10.00 MMHOS/CM. Some general guidelines are:"
1550 PRINT:PRINT TAB(12) "Soil Salinity Class Conductivity Reading"
1555 PRINT TAB(12) "-----"
1560 PRINT TAB(12) "Non - saline 0.00 - 0.99"

```

```

1565 PRINT TAB(12) "Very slightly saline           1.00 - 1.99"
1570 PRINT TAB(12) "Slightly saline                 2.00 - 3.99"
1575 PRINT TAB(12) "Moderately saline              4.00 - 6.99"
1580 PRINT TAB(12) "Strongly saline                 over  7.00"
1581 PRINT:PRINT TAB(7) "Soil salinity estimates can be obtained from
      tables giving physical"
1582 PRINT TAB(7) "and chemical properties of soils in detailed soil
      survey publications."
1585 LOCATE 25,23: PRINT "press any key to continue";
1590 A$ = INKEY$: IF A$ = "" THEN 1590
1595 CLS:LOCATE 25,10,1: PRINT FKEY1$;:RETURN
1600 ^-----> help statement for determining normal yield
1605 CLS: LOCATE 3,12,0: PRINT" Normal yield estimates can be your own
      estimates based on"
1610 PRINT TAB(7) "experience or they can be based on variety performance
      data from branch"
1615 PRINT TAB(7) "experiment stations. They can also be based on soil
      survey reports."
1620 LOCATE 24,21: PRINT "press any key to continue"
1625 A$ = INKEY$: IF A$ = "" THEN 1625
1630 RETURN
1635 ^-----> help statement for field identification
1640 CLS: LOCATE 3,12,0: PRINT "Your field identification can be any number
      or letter or"
1645 PRINT TAB(7) "any combination such as a word or a sentence. It can be up"
1650 PRINT TAB(7) "to 25 characters in length."
1655 LOCATE 25,23: PRINT "press any key to continue";
1660 A$ = INKEY$: IF A$ = "" THEN 1660
1665 LOCATE 25,10,1: PRINT FKEY1$;:RETURN
1670 ^-----> help for choices
1675 LOCATE 20,10,0: PRINT "Currently you are limited to the seven choices
      displayed"
1680 PRINT TAB(7) "in the menu. Simply press the number of the crop you
      want to see"
1685 PRINT TAB(7) "and you will be prompted for further information.";
1690 FOR I% = 1 TO 5000: NEXT I%
1695 LOCATE 20,1,0
1700 FOR I% = 1 TO 3
1705 PRINT STRING$(79,32)
1710 NEXT I%
1715 RETURN
1720 ^-----> help for yields
1725 LOCATE 20,10,0: PRINT "Enter the yield you normally get or anticipate
      getting"
1730 PRINT TAB(7) "for the field you named. Enter only the digits and
      round to the "
1735 PRINT TAB(7) "nearest whole number."
1740 FOR I% = 1 TO 5000: NEXT I%
1745 LOCATE 20,1: FOR I% = 1 TO 3: PRINT ST$;:NEXT I%:LOCATE 25,10,1:
      PRINT FKEY2$;:RETURN
1750 ^-----> provide detailed comments
1755 N% = 1: DIM RECORD$(11)
1760 LOCATE 20,1:PRINT ST$;:LOCATE 20,7:PRINT "Do you want a hardcopy of
      the comments(Y/N)?";
1765 CALLER = 5: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 2
1770 GOSUB 2290

```

```

1775 IF S$ = "y" OR S$ = "Y" THEN LP% = -1 ELSE LP% = 0
1780 LOCATE 20,1:PRINT ST$: LOCATE 20,7: PRINT "What number([";CR$;"] to stop)";
1785 CALLER = 2: XFLD = POS(0)+1: YFLD = CSRLIN: LNGTH = 3
1790 GOSUB 2290
1795 IF S$ = "" THEN 1820
1800 RECORD$(N%) = S$
1805 LOCATE 3+VAL(S$),9: PRINT CHR$(251);
1810 N% = N% + 1
1815 GOTO 1780
1820 CLS:FOR X% = 1 TO N%-1
1825 READ CMT$
1830 WHILE CMT$ <> "bquit"
1835 IF CMT$<>RECORD$(X%) THEN 1855
1840 READ CMT$
1845 PRINT TAB(12) CMT$
1850 IF LP% = -1 THEN LPRINT CMT$
1855 READ CMT$
1860 WEND
1865 RESTORE
1870 NEXT X%
1875 LOCATE 25,21: PRINT "press any key to continue";
1880 A$ = INKEY$: IF A$ = "" THEN 1880
1885 RETURN
1890 '-----> printer routine
1895 LOCATE 25,1: PRINT ST$;
1900 LOCATE 25,20: PRINT "For copy of screen press [";CHR$(24);"]
and [PrtSc] keys";
1905 LPRINT CHR$(27);CHR$(53)
1910 LPRINT CHR$(14);" Soil Salinity is ";SS$;" MMHOS/cm"
1915 LPRINT
1920 LPRINT DATE$" at "TIME$ TAB(45) "Field: ";FIELDID$
1925 LPRINT
1930 FOR X% = 1 TO (NUM%-1)
1935 LPRINT "Crop: ";CROPNAME$(X%)
1940 LPRINT "Estimated Normal Yield: ";NRMLYLD(X%)
1945 LPRINT "Expected Final Yield: ";: LPRINT USING "####.#";FINALYLD(X%)
1950 LPRINT
1955 NEXT X%
1960 LPRINT CHR$(12)
1965 GOTO 530
1970 '*****
1975 '* EXPANDED COMMENTS *
1980 '* The following are expanded comments for the Summary Comments listed*
1985 '* earlier in the program. *
1990 '*****
1995 DATA "1","1. Identify the source of the salinity problem. Soil test,"
2000 DATA "1","refer to publications and seek professional advice for"
2005 DATA "1","diagnosis and interpretation."
2010 DATA "1"," "
2015 DATA "2","2. Investigate the possibility of improvement by surface and/"
2020 DATA "2","or subsurface drainage."
2025 DATA "2"," "
2030 DATA "3","3. Use deep rooted forage crops to remove some of the excess"
2035 DATA "3","water in seep situations. Seeded above saline seep discharge"
2040 DATA "3","areas, alfalfa and wild ryegrass serve well as groundwater"
2045 DATA "3","interceptor crops."

```

2050 DATA "3", " "
 2055 DATA "4", "4. Use salt-tolerant crops. Six-row barley is the most salt-"
 2060 DATA "4", "tolerant cereal crop. Of the oilseeds, sunflower and safflower"
 2065 DATA "4", "perform well. Forages, of course, are the ultimate crops for"
 2070 DATA "4", "seeding in saline areas. It is recommended that mixtures of"
 2075 DATA "4", "forage crops rather than single species be used in seeding salt-"
 2080 DATA "4", "affected soils."
 2085 DATA "4", " "
 2090 DATA "5", "5. Seed to take advantage of moisture conditions. Late fall"
 2095 DATA "5", "seeding of forage crops usually allows the best moisture"
 2100 DATA "5", "situations for germination in the spring. Seeding forages"
 2105 DATA "5", "in the spring period immediately after rain should be advantageous"
 2110 DATA "5", " "
 2115 DATA "6", "6. Sweet clover used in rotation as a hay crop or for green"
 2120 DATA "6", "manure adds nitrogen and organic matter to the soil, thereby"
 2125 DATA "6", "improving water movement and tilth."
 2130 DATA "6", " "
 2135 DATA "7", "7. Barnyard manure applied at 15 to 25 tons per acre every"
 2140 DATA "7", "three to four years can help reclaim moderately saline soils."
 2145 DATA "7", "Applied as a light mulch before seeding, manure holds moisture"
 2150 DATA "7", "near the soil surface and enhances crop germination."
 2155 DATA "7", " "
 2160 DATA "8", "8. Fertilizers, used as recommended by soil test, help to boost"
 2165 DATA "8", "production and utilize more water, and should be a part of overall"
 2170 DATA "8", "planning in managing saline soils. Refer to Extension circulars"
 2175 DATA "8", "SF-713, `Fertilizing Sunflowers`; SF-722, `Fertilizing Corn Grain"
 2180 DATA "8", "and Silage; SF-712, `Fertilizing Wheat, Durum & Rye`; SF-723,"
 2185 DATA "8", "`Fertilizing Malting and Feed Barley`; or SF-717, `Fertilizing"
 2190 DATA "8", "Flax` for further information."
 2195 DATA "8", " "
 2200 DATA "9", "9. Since summerfallowing in recharge areas appears to contribute"
 2205 DATA "9", "to increased salinity, the practice should be avoided whenever"
 2210 DATA "9", "possible. In saline areas minimum tillage should be practiced."
 2215 DATA "9", "Any growth, even weeds, is helpful in drying out saline seeps."
 2220 DATA "9", "Weeds can be prevented from going to seed by timely herbicide"
 2225 DATA "9", "application. The concept of zero tillage may be advantageous"
 2230 DATA "9", "under saline conditions."
 2235 DATA "9", " "
 2240 DATA "10", "10. Cultivate when the soil is neither too wet nor too dry."
 2245 DATA "10", "Deep plowing to mix the lime and gypsum with soil layers high"
 2250 DATA "10", "in sodium has proved beneficial on solonchic(alkali) soils."
 2255 DATA "10", "Remember, though, deep plowing is not a management or reclamation"
 2260 DATA "10", "practice for saline salts."
 2265 DATA "10", " "
 2270 DATA "11", "11. There are no chemical amendments which will `neutralize`"
 2275 DATA "11", "the salts present in saline soils."
 2280 DATA "11", " "
 2285 DATA "bquit"
 2290 `-----> screenin routine
 2295 `-----> taken from `Advanced Basic and Beyond` by Dr. L. Goldstein
 2300 `-----> this routine inputs data as a string S\$
 2305 `It allows input to have the following parameters:
 2310 `LNGTH = maximum length of input string
 2315 `XFLD = cursor column for beginning of input string
 2320 `YFLD = cursor row for input field
 2325 `FLDBEG = first character position in field

```

2330 ^FLDEND = last character position in field
2335 ^S$ = contents of the field from beginning up to space before cursor
2340 ^T$ = contents of the field from the cursor to the end of the field
2345 ^At the end of the routine , the contents of the field are returned in S$
2350 ^LASTPOS = position currently occupied by last character
2355 ^If a key with an extended ASCII code is pressed, it ends processing the
2360 ^current field
2365 ^The contents of the field are returned in S$, the second byte of the
2370 ^extended ASCII code in E$.
2375 ^CSR = the current column of the cursor
2380 ^-----> main routine
2385 S$ = "": E$ = "": INPUTEND = FALSE: KEYHIT = FALSE
2390 FLDEND = XFLD + (LNGTH-1)
2395 CSR = XFLD
2400 GOSUB 2720 ^-----> compute initial lastpos
2405 LOCATE YFLD, XFLD: COLOR 0,7:PRINT SPC(LNGTH-1):COLOR 7,0:LOCATE YFLD,XFLD
2410 WHILE INPUTEND = FALSE
2415 GOSUB 2800 ^-----> input character
2420 IF EXTENDED = TRUE THEN 2490 ELSE 2465
2425 WEND
2430 J% = (XFLD+LNGTH) - LASTPOS
2435 PRINT STRING$(J%,32);
2440 GOSUB 2640 ^-----> read screen
2445 S$ = S$ + T$
2450 RETURN
2455 ^-----> begin subroutines <-----
2460 ^-----> handle ordinary ASCII codes
2465 KEYHIT = TRUE
2470 IF C$ = CHR$(8) THEN 2530 ^Backspace
2475 IF C$ = CHR$(13) THEN 2565 ^Enter
2480 IF C$ = CHR$(27) THEN 2580 ^Esc
2485 IF C$ >= CHR$(32) THEN GOTO 2610 ^handle displayable character
2490 ^-----> handle extended ASCII codes
2495 E$ = C$
2500 INPUTEND = TRUE
2505 GOTO 2425
2510 ^-----> reject character
2515 BEEP
2520 GOTO 2425
2525 ^-----> handle backspace
2530 IF LASTPOS < XFLD THEN 2515
2535 GOSUB 2640
2540 IF CSR <> XFLD THEN CSR = CSR-1:PRINT CHR$(219);
2545 LOCATE YFLD,CSR:PRINT CHR$(219);:LOCATE YFLD,CSR
2550 LASTPOS = LASTPOS-1
2555 GOTO 2425
2560 ^-----> handle enter
2565 INPUTEND = TRUE
2570 GOTO 2425
2575 ^-----> handle esc
2580 LOCATE YFLD,XFLD
2585 PRINT STRING$(LNGTH,32);
2590 LASTPOS = 0: CSR = XFLD
2595 LOCATE YFLD, XFLD:COLOR 0,7:PRINT SPC(LNGTH-1):COLOR 7,0:LOCATE YFLD,XFLD
2600 GOTO 2425
2605 ^-----> display character

```

```

2610 PRINT C$;
2615 IF LASTPOS < CSR THEN LASTPOS = CSR
2620 IF CSR = FLDEND THEN BEEP
2625 IF CSR = FLDEND THEN PRINT CHR$(29); CHR$(32); CHR$(29);ELSE CSR = CSR + 1
2630 GOTO 2425
2635 ^-----> read field from screen
2640 LOCATE,,0
2645 S$ = ""; T$ = ""
2650 IF LASTPOS = 0 THEN 2690
2655 FOR J% = XFLD TO CSR-1
2660   S$ = S$ + CHR$(SCREEN(YFLD,J%))
2665 NEXT J%
2670 FOR J% = CSR TO LASTPOS
2675   T$ = T$ + CHR$(SCREEN(YFLD,J%))
2680 NEXT J%
2685 LOCATE,,1
2690 RETURN
2695 ^-----> erase field
2700 LOCATE YFLD, XFLD: CSR = XFLD: LASTPOS = 0
2705 PRINT STRING$(LNTH,32);
2710 LOCATE YFLD, XFLD
2715 RETURN
2720 ^-----> compute lastpos
2725 LASTPOS = FLDEND: CSR = XFLD
2730 GOSUB 2640
2735 WHILE RIGHT$(T$,1) = CHR$(32)
2740   T$ = LEFT$(T$,LEN(T$)-1)
2745   LASTPOS = LASTPOS -1
2750 WEND
2755 RETURN
2760 ^-----> clear keyboard buffer
2765 DEF SEG = 0: POKE 1050,PEEK(1052)
2770 DEF SEG
2775 RETURN
2780 ^-----> end routines
2785 ^-----> keyin
2790 ^This routine reads a character from the keyboard and accepts or
2795 ^rejects it based on the caller's specifications
2800 ^-----> input character string from inkey$
2805 C$ = INKEY$
2810 IF C$ = "" THEN 2805
2815 C = ASC(C$)
2820 IF LEN(C$) = 2 THEN EXTENDED = TRUE ELSE EXTENDED = FALSE
2825 IF EXTENDED = FALSE THEN 2855
2830 C$ = RIGHT$(C$,1)
2835 C = ASC(C$)
2840 GOTO 2910
2845 ^-----> ordinary ASCII codes
2850 ^----> test for range
2855 IF C >= MINKEY(CALLER) AND C <= MAXKEY(CALLER) THEN 2880
2860 ^-----> handle special characters
2865 IF SPECIALKEY$(CALLER) = "" THEN 2940
2870 IF INSTR(SPECIALKEY$(CALLER),C$) = 0 THEN 2940
2875 ^-----> convert to capitals if necessary
2880 IF CAPSON(CALLER) = FALSE THEN 2950
2885 IF C > 96 AND C < 123 THEN C = C-32

```

```

2890 C$ = CHR$(C)
2895 GOTO 2950
2900 '-----> extended ASCII codes
2905 '----> test for range
2910 IF C >= EXTMINKEY(CALLER) AND C <= EXTMAXKEY(CALLER) THEN 2950
2915 '-----> handle special characters
2920 IF EXTSPECIALKEY$(CALLER) = "" THEN 2940
2925 IF INSTR(EXTSPECIALKEY$(CALLER),C$) = 0 THEN 2940
2930 GOTO 2950
2935 '-----> illegal character
2940 BEEP
2945 GOTO 2805
2950 RETURN
2955 '-----> ERROR TRAPPING
2960 IF ERR = 25 THEN 2965 ELSE 2980
2965 CLS: LOCATE 3,20:PRINT "Your printer is not on!"
2970 PRINT TAB(10) "Make sure to align your forms to top-of-page."
2975 GOTO 3005
2980 CLS: LOCATE 3,10: PRINT "An error has occurred in the program that is not"
2985 PRINT TAB(10) "recoverable. Type `RUN` to restart the program."
2990 END
2995 '
3000 '
3005 FOR I% = 1 TO 5000: NEXT I%: RESUME 525

```